

**Собственные числа
и собственные вектора матрицы**

$$A := \begin{pmatrix} 1.342 & 0.432 & -0.599 & 0.202 \\ 0.432 & 2.145 & 0.256 & -0.599 \\ -0.599 & 0.256 & 0.382 & 0.532 \\ 0.202 & -0.599 & 0.532 & 3.047 \end{pmatrix}$$

Использование функций MathCAD

L := eigenvals ((A))

W := eigenvecs (A)

Метод вращений.

$$f(i, j, A) := \text{if} \left[\left(A_{i,i} \neq A_{j,j} \right), 0.5 \operatorname{atan} \left(\frac{2 \cdot A_{i,j}}{A_{i,i} - A_{j,j}} \right), \frac{\pi}{4} \right]$$

c(i, j, A) := cos (f(i, j, A))

s(i, j, A) := sin(f(i, j, A))

$$t(i, j, A) := \sum_{i=0}^3 \sum_{j=0}^3 \left[(A_{i,j})^2 \right] - \sum_{i=0}^3 (A_{i,i})^2$$

**i - номер строки и j - номер столбца
максимального по модулю элемента матрицы A,
расположенного выше главной диагонали**

Итерация 1

V := identity (4) i := 0 j := 2

U := identity (4)

U_{i,i} := c(i, j, A) U_{j,j} := c(i, j, A) U_{i,j} := -s(i, j, A) U_{j,i} := s(i, j, A)

A := U^T · A · U V := V · U

t(i, j, A) = 1.87

$$A = \begin{pmatrix} 1.63 & 0.279 & 0 & -0.048 \\ 0.279 & 2.145 & 0.418 & -0.599 \\ 0 & 0.418 & 0.094 & 0.567 \\ -0.048 & -0.599 & 0.567 & 3.047 \end{pmatrix}$$

Итерация 2

i := 1 j := 3

U := identity (4)

U_{i,i} := c(i, j, A) U_{j,j} := c(i, j, A) U_{i,j} := -s(i, j, A) U_{j,i} := s(i, j, A)

A := U^T · A · U V := V · U

$$t(i, j, A) = 1.152$$

$$A = \begin{pmatrix} 1.63 & 0.228 & 0 & -0.167 \\ 0.228 & 1.846 & 0.627 & 0 \\ 0 & 0.627 & 0.094 & 0.321 \\ -0.167 & 0 & 0.321 & 3.346 \end{pmatrix}$$

И так далее...

Степенной метод.

$$n := 50$$

$$y^{(0)} := \begin{pmatrix} 1 \\ 1 \\ 1 \\ 1 \end{pmatrix}$$

$$k := 0..n$$

$$y^{(k+1)} := A \cdot y^{(k)}$$

$$\lambda_1 := \frac{(y^{(n+1)})_0}{(y^{(n)})_0}$$

Сравнение

$$L = \begin{pmatrix} -0.135380724 \\ 1.517658394 \\ 2.139706618 \\ 3.394015711 \end{pmatrix}$$

$$\lambda_1 = 3.394015715$$

$$y^{(n+1)} := \frac{y^{(n+1)}}{|y^{(n+1)}|}$$

Невязка

$$A \cdot y^{(n+1)} - \lambda_1 \cdot y^{(n+1)} = \begin{pmatrix} 6.276 \times 10^{-11} \\ 1.709 \times 10^{-9} \\ -6.072 \times 10^{-10} \\ -4.153 \times 10^{-9} \end{pmatrix}$$